


Application Note		資料作成 100324	資料番号 an2k-019
テーマ	XY 軸円弧補間と Z 軸を同期させて螺旋を描く		
使用機器	MPC-1000/2000/2100、MPG-2314、MPC-XY03、		

■機器構成



■概要

- XY 軸円弧補間と Z 軸を組み合わせ螺旋を描きます。
- MPG-2314 の円弧補間は 3 軸の同期が取れないので、Z 軸の動作速度を移動距離(パルス数)と旋回所要時間から算出して擬似的にシンクロさせています。
- 旋回 1~10 回転の所要時間は予め測定してタイムテーブルに入力してあります。
- DSW で旋回数を設定します(0=10 回)
- このサンプルの速度(ACCEL 設定値)、半径は固定です。

■プログラム

```

DIM XYT(11)
/* タイムテーブル。XY 軸旋回時間の SYSCLK 実測値 (mSec)
XYT(1)=2918 /* 1 回転
XYT(2)=5746
XYT(3)=8574
XYT(4)=11401
XYT(5)=14229
XYT(6)=17056
XYT(7)=19884
XYT(8)=22711
XYT(9)=25540

```

```

XYT(10)=28367

PG 0 /* MPG-2314 1 枚目

DO
  GOSUB *HOME_Z
  GOSUB *HOME_XY
  GOSUB *SPIRAL
LOOP

*SPIRAL

ACCEL ALL_A 30000 1000 500
MOVL 10000 0 VOID -11000 /* 作業開始点へ移動
WAIT RR(ALL_A)==0
STPS 0 0 VOID VOID /* X,Y 軸ここを'0'にセット
TIME 100

REVO=IN(24)/&H10 /* XY 回転数(DSW 読み込み)
IF REVO==0 THEN : REVO=10 : END_IF
PRINT "回転数" REVO

ZL=10000 /* Z 軸目標パルス数
ZS=ZL*1000/XYT(REVO) /* Z 軸速度計算

Aset=X_A|Y_A /* 軸指定
ACCEL Aset 20000 2000 1000 /* 加速有り
/*ACCEL Aset 20000 1 20000 /* 加速無し。この場合のタイムテーブルは上記と異なる
ACCEL Z_A ZS 1 ZS /* Z 軸 最高速度=最低速度 つまり加減速無し
SYSCLK=0
RMVS Z_A ZL+1000 /* Z 軸上昇 目標パルス数より多く設定
/*DS_DACL /* 減速無効 注)円弧補間の停止は即停止のみ
FOR I=1 TO REVO
  RMVT Aset 0 0 CCW 0 10000 /* 現在点から相対座標 0,0 へ 0,10000 を中心として CCW 移動
NEXT
/*EN_DACL /* 減速有効。連続補間の最後が直線で減速停止する場合使う

WAIT RR(Aset)==0 /* XY 動作完了待ち
STOP Z_A STP_I /* Z 軸即停止
WAIT RR(Z_A)==0 /* Z 軸停止待ち

PR "時間 Z 座標" SYSCLK Z(0)
TIME 500
RETURN

' Z 軸原点復帰
*HOME_Z
ACCEL Z_A 10000 1 10000 /* 原点復帰スピード
IF HPT(ZIN0)<>0 THEN
  RMVL 0 0 0 -5000
  WAIT RR(ALL_A)==0
  TIME 100
END_IF
SHOM Z_A INO_ON
HOME 0 0 0 50000
WAIT RR(ALL_A)==0
TIME 100
RMVS Z_A -500 /* 必要に応じてオフセット
WAIT RR(ALL_A)==0
STPS Z_A 0 /* Z 軸ここを'0'にセット
/*PRINT "Z HOME"
RETURN

' XY 軸原点復帰
*HOME_XY
ACCEL X_A|Y_A 10000 1 10000 /* 原点復帰スピード
IF HPT(XIN0)<>0 THEN /* X 軸 IN0 がオンなら退避移動
  RMVS X_A 10000
END_IF
IF HPT(YIN0)<>0 THEN /* Y 軸 IN0 がオンなら退避移動
  RMVS Y_A 10000
END_IF
WAIT RR(ALL_A)==0
SHOM X_A|Y_A INO_ON
HOME -100000 -100000 0 0
WAIT RR(ALL_A)==0

```

```
TIME 100
RMVL 500 500 0 0          /* オフセット
WAIT RR(ALL_A)==0
STPS 0 0 VOID VOID      /* X,Y軸ここを'0'にセット
/*PRINT "XY HOME"
TIME 100
RETURN
```

実行結果

```
回転数 1
時間 Z座標 2918 -995 ← Zの目標値は -1000、この場合速度演算の分解能による誤差などで5パルス多い。
回転数 2
時間 Z座標 5746 -998
回転数 3
時間 Z座標 8574 -999
回転数 4
時間 Z座標 11402 -997
回転数 5
時間 Z座標 14230 -1007
回数 6
時間 Z座標 17057 -1001
回転数 7
時間 Z座標 19885 -1014
回転数 8
時間 Z座標 22712 -1003
回転数 9
時間 Z座標 25540 -1010
回転数 10
時間 Z座標 28367 -1011
```

以上